

EESy Solutions

Engineering Equation Solver Newsletter

Inside this issue:

Welcome	1
RealThermoProps	1
The Variable Information Window	4
\$VarInfo Directive	6
Mixture Properties	7
Additional Changes to EES	8
EES Instant Update	8

Welcome

This is the 34th issue of EESy Solutions, a newsletter that provides news, tips, and other updates for users of the Engineering Equation Solver software. This issue focuses on new property functions, fluid mixtures, and Variable Information.

EES has been commercially available for more than two decades. If you have missed any of the previous issues of EESy Solutions, they can be downloaded from www.fchart.com.

RealThermoProps

In the process of simulating any thermodynamic cycle it is usually necessary to obtain a complete set of thermodynamic properties at each state point. The resulting exercise of writing EES code can be tedious, as each property requires a separate call to a unique function in EES' internal property database (e.g., Enthalpy, Entropy, Volume, Quality, Temperature, Pressure, etc.).

RealThermoProps provides a coding shortcut that allows you to determine all of the properties at a given state in a single call. The procedure will return temperature, pressure, specific volume, specific enthalpy, specific entropy, specific internal energy, and quality (in that order) for a specified fluid and state.

The procedure is called in same way as a typical thermodynamic property functions in EES, requiring the name of the fluid and two arguments with property identifiers that fix the state. The difference is that the procedure will return each of the outputs listed above.

A simple example is below:

```


$UnitSystem SI C kPa mass
T=25 [C]
P=100 [kPa]
R$='R134a'
Call RealThermoProps(R$, T=T, P=P : T, P ,v ,h ,s ,u ,x)
    
```

Note that T and P in the argument list already have values and therefore are not re-assigned. The remaining properties are evaluated.

RealThermoProps (continued)

Consider a simple vapor compression system. The refrigerant, evaporating temperature, condensing temperature, and compressor efficiency are all given. There is assumed to be no superheat or subcool and no pressure loss in the evaporator and condenser.

An EES program that determines the properties at each of the four states and calculates the Coefficient of Performance (COP) of the cycle with out using the RealThermoProps function is shown below.



```
$UnitSystem SI Mass kJ C kPa
```

```
T_cond=30 [C]
```

```
T_evap= -5 [C]
```

```
F$=R410a'
```

```
eta_c=0.65 [-]
```

```
"condensing temperature"
```

```
"evaporating temperature"
```

```
"refrigerant"
```

```
"compressor efficiency"
```

```
"Evaporator outlet"
```

```
T[2]=T_evap
```

```
x[2]=1
```

```
P[2]=Pressure(F$, T=T[2], x=x[2])
```

```
s[2]=Entropy(F$, T=T[2], x=x[2])
```

```
h[2]=Enthalpy(F$, T=T[2], x=x[2])
```

```
v[2]=Volume(F$, T=T[2], x=x[2])
```

```
"Condenser outlet"
```

```
T[4]=T_cond
```

```
x[4]=0
```

```
P[4]=Pressure(F$, T=T[4], x=x[4])
```

```
s[4]=Entropy(F$, T=T[4], x=x[4])
```

```
h[4]=Enthalpy(F$, T=T[4], x=x[4])
```

```
v[4]=Volume(F$, T=T[4], x=x[4])
```

```
"Compressor outlet"
```

```
P[3]=P[4]
```

```
s_3s=s[2]
```

```
h_3s=Enthalpy(F$, P=P[3], s=s_3s)
```

```
h[3]=h[2]+(h_3s-h[2])/eta_c
```

```
T[3]=Temperature(F$, h=h[3], P=P[3])
```

```
s[3]=Entropy(F$, h=h[3], P=P[3])
```

```
v[3]=Volume(F$, h=h[3], P=P[3])
```

```
"Valve outlet"
```

```
P[1]=P[2]
```

```
h[1]=h[4]
```


```
T[1]=Temperature(F$, h=h[1], P=P[1])
```

```
s[1]=Entropy(F$, h=h[1], P=P[1])
```

```
v[1]=Volume(F$, h=h[1], P=P[1])
```

```
"coefficient of performance"
```

```
COP=(h[2]-h[1])/(h[3]-h[2])
```



RealThermoProps (continued)

Consider the same program but this time using the RealThermoProps procedure.

```

$UnitSystem SI Mass kJ C kPa
T_cond=30 [C]      "condensing temperature"
T_evap=-5 [C]     "evaporating temperature"
F$='R410a'        "refrigerant"
eta_c=0.65 [-]    "compressor efficiency"

"Evaporator outlet"
T[2]=T_evap; x[2]=1
Call RealThermoProps(F$, T=T[2], x=x[2]: T[2], P[2], v[2], h[2], s[2], u[2], x[2])

"Condenser outlet"
T[4]=T_cond; x[4]=0
Call RealThermoProps(F$, T=T[4], x=x[4]: T[4], P[4], v[4], h[4], s[4], u[4], x[4])

"Compressor outlet"
P[3]=P[4]
s_3s=s[2]
h_3s=Enthalpy(F$, P=P[3], s=s_3s)
h[3]=h[2]+(h_3s-h[2])/eta_c
Call RealThermoProps(F$, h=h[3], P=P[3]: T[3], P[3], v[3], h[3], s[3], u[3], x[3])

"Valve outlet"
P[1]=P[2]; h[1]=h[4]
Call RealThermoProps(F$, P=P[1], h=h[1]: T[1], P[1], v[1], h[1], s[1], u[1], x[1])

COP=(h[2]-h[1])/(h[3]-h[2])    "coefficient of performance"

```

Unwanted outputs from the procedure can be skipped by leaving those spots in the output list blank, as shown below. Note how much less typing is required when RealThermoProps is used.

```

$UnitSystem SI Mass kJ C kPa
T_cond=30 [C]      "condensing temperature"
T_evap=-5 [C]     "evaporating temperature"
F$='R410a'        "refrigerant"
eta_c=0.65 [-]    "compressor efficiency"

"Evaporator outlet"
T[2]=T_evap; x[2]=1
Call RealThermoProps(F$, T=T[2], x=x[2]: , P[2], v[2], h[2], s[2], , )

"Condenser outlet"
T[4]=T_cond; x[4]=0
Call RealThermoProps(F$, T=T[4], x=x[4]: , P[4], v[4], h[4], s[4], , )

"Compressor outlet"
P[3]=P[4]
s_3s=s[2]
h_3s=Enthalpy(F$, P=P[3], s=s_3s)
h[3]=h[2]+(h_3s-h[2])/eta_c
Call RealThermoProps(F$, h=h[3], P=P[3]: T[3], , v[3], h[3], s[3], , )



"Valve outlet"
P[1]=P[2]; h[1]=h[4]
Call RealThermoProps(F$, P=P[1], h=h[1]: T[1], , v[1], , s[1], , x[1])

COP=(h[2]-h[1])/(h[3]-h[2])    "coefficient of performance"


```

The Variable Information Window

The Variable Information Window provides a table where the information about each of the variables that are used in the Equations Window can be found. For example, open the example file CapVsT.ees (select Properties, thermodynamic from the Examples menu and then A heat pump cycle using R134a).




The Variable Information Window for the program is shown below. The name of each variable is listed together with its guess value and limits. In addition there are options that allow you to specify how the variable is displayed. The units of each variable can be set and the alternate units can also be set. Finally, the variable can be designated to be a key variable and comments can be set that will appear in the Key Variable Window in the Solution.




The Variable Information Window (continued)


Suppose that the variable VolFlow (which represents the volumetric flow rate at the suction of the compressor) were of particular interest. The variable could be designated to have alternate units (e.g., liter/min) and be a key variable, as shown below.



The Solution Window would now contain a Key Variables tab.



Another useful feature of the Variable Information Window is the ability to quickly change the name of a variable globally. For example, suppose that it was desired to change the name of VolFlow to V_dot_suction. Select the name of the variable VolFlow in the Variable Information Window and change its name to V_dot_suction. Select OK and you will be prompted with the dialog shown below.




Select OK and you will find that the variable VolFlow has been changed to V_dot_suction everywhere in the program. This includes not only the Equations Window but also Parametric Tables, the Diagram Window, etc. Further, the variable V_dot_suction retains all of the same characteristics (guess value, limits, units, display, etc.) that were originally assigned to variable.

The \$VarInfo Directive

The \$VarInfo directive allows the same information that is entered in the Variable Information Window to be entered in the Equations Window. The format for the \$VarInfo directive is:

```
$VarInfo VariableName Guess=V1 Lower=V2 Upper=V3 Units='XXXX' AltUnits='YYYY' Display=F1
Style=Bold Key='Key variable comment'
```

As an example, see the specifications for variable VolFlow in the Variable Information Window below:



The same variable specifications can also be accomplished using the \$VarInfo directive:

```
$VarInfo VolFlow Guess = 0.0043 Units = 'm^3/s' AltUnits = 'liter/min' Key = 'Volumetric flow rate at the compressor suction'
```

The \$VarInfo directive is particularly useful to set the characteristics of all of the elements in an array that might change size. For example, the program below utilizes the array T[] which has size N where N is a variable. The use of the \$VarInfo directive followed by T[] sets the characteristics (in this case the units) of all of the elements of the array T, regardless of its size upon execution. \$VarInfo can be used within Functions and Procedures, as well as in the Main program.

```
Procedure Calct(N:T[1..100])
$VarInfo T[] Units='K'
T[1]=300 [K]
Duplicate i=2,N
T[i]=T[i-1]+1 [K]
End
End


$VarInfo T[] Units='K'
N=10
Call Calct(N:T[1..N])
```

The screenshot shows a software interface with a 'Main' window. It features a 'Sort' button and a table with 10 rows representing the elements of the array T[]:

	T _i
[1]	300 [K]
[2]	301
[3]	302
[4]	303
[5]	304
[6]	305
[7]	306
[8]	307
[9]	308
[10]	309

Zeotropic Mixtures

Thermodynamic and transport property information for several zeotropic mixtures have been added to EES. For example, property plots for R452A and R454C are shown below; notice the temperature glide that is evident within the vapor dome.



Other refrigerant mixtures that are now provided in the EES database are: R448A, R449A, R450A, R452B, R513A, and R514A.

Additional Changes to EES

- The \$ClearParametric directive allows a specific set of rows in a Parametric Table to be cleared before calculations are initiated.
- The \$Reference directive will accept a string variable for the fluid name.
- The RetrieveParametric macro allows a .txt or .par file containing a Parametric Table to be opened.
- If an alternate unit has been specified for a variable in a Parametric or Lookup Table, then the Professional version allows values to be entered in the Alter Values or the Change Column Values dialog in either unit set. Both X-Y and X-Y-Z plots can be constructed using either the base or alternate unit set.
- The Paste Special button appears in the upper left cell of Lookup Tables when data is available in the clipboard. The button allows data to be pasted into a Lookup table starting at a specific row or column. Column names and units may also be pasted from the clipboard.
- Properties for Steam, Water, R718, and Steam_IAPWS now all use the Steam_IAPWS property correlations issued by the International Association for the Properties of Water and Steam, which provide the highest accuracy.
- The \$Input directive in the Professional version provides a dialog box to enter the value of an EES variable when calculations are initiated
- The \$CopytoLookup directive allows the units of the column to which the data are being copied to be set.
- The Align tool on the Diagram Window toolbar can be used to align buttons as well as text and objects.
- The NewPlot and OverlayPlot macro commands will accept a PointLabels keyword which creates labels that identify each plotted point with its index or row.



F-Chart Software

PO Box 44042
Madison, WI, 53744

Phone/FAX: 608-274-4262
Internet: <http://fchart.com>
E-mail: info@fchart.com

Instant Update and Technical Service

EES uses a different model for updating than most other programs. Each time that there is a change in the EES program, either to correct a problem or to add a new feature, the version number is incremented by 0.001 and the latest version of EES is placed on our website. Although the program has become very robust and stable, there have been many new versions with new features released since the last EESy Solutions was distributed. A list of new features is maintained at:

<http://fchart.com/ees/new-features.php>.

Any user who has a current subscription to Instant Update and Technical Service (IUTS) can download the latest version. All new licenses of EES are provided with one year of IUTS. The fee to continue IUTS after the first year is 20% of the current cost of the program per year, provided that it is renewed within 12 months after expiration of the Service. Contact F-Chart Software if you wish to re-subscribe to Instant Update and Technical Service.